

## Limits and Continuity

```
> restart:with(plots):  
> setoptions3d(axes=NORMAL,labels=["x","y","z"],orientation=[20,  
70]);
```

Maple is able to compute some limits, such as  $\lim_{(x,y) \rightarrow (3,2)} \frac{3xy^2 + 6xy}{2x^2y + 4y}$ .

```
> limit((3*x*y^2+6*x*y)/(2*x^2*y+4*y),{x=3,y=2});  
18  
11
```

But other limits it cannot compute.

```
> limit(sin(x^2+y^2)/(x^2+y^2),{x=0,y=0});  
limit( $\frac{\sin(x^2 + y^2)}{x^2 + y^2}, \{x=0, y=0\}$ )
```


That includes instances where the limit does not exist.

```
> f:=(2*x^2*y)/(x^4+y^2);  
> limit(f,{x=0,y=0});
```

$$f := \frac{2x^2y}{x^4 + y^2}$$
$$\text{limit}\left(\frac{2x^2y}{x^4 + y^2}, \{x=0, y=0\}\right)$$


The density plot below, with the parabolas  $y = x^2$  and  $y = -x^2$  superimposed, shows the lack of a limit at  $(0,0)$ . Notice the high value of **numpoints**.

```
> p1:=densityplot(f,x=-9..9,y=-9..9,style=patchnogrid,color=green,  
numpoints=10000):  
p2:=plot(x^2,x=-9..9,y=-9..9):  
p3:=plot(-x^2,x=-9..9,y=-9..9):  
display(p1,p2,p3);
```



We see that approaching (0,0) along the lower parabola results in higher values than approaching (0,0) along the upper parabola. Now let's look at the graph of the function with the two parabolic paths superimposed.

```
> p1:=plot3d(f,x=-9..9,y=-9..9,style=patchnogrid,numpoints=10000):  
p2:=spacecurve([t,t^2,1],t=-2..2,color=red):  
p3:=spacecurve([t,-t^2,-1],t=-2..2,color=blue):  
display(p1,p2,p3);
```



We see clearly that there is no limit at  $(0,0)$ . Thus this function is not continuous at  $(0,0)$ , although it is continuous everywhere else.

>